

## Chapter 6

# Predicate logic

Predicate logic is the language of Mathematics. Unlike propositional logic which on its own isn't good for anything we can express any mathematical concept in predicate logic. This is why predicate logic has been used to by Peano to present the theory of natural numbers and later by Zermelo and Fraenkel to write down the axioms of set theory. However, from the type-theoretic perspective we don't have to present the formalism of predicate logic but as for propositional logic we just need to explain how we can represent the idioms of predicate logic in Type Theory using the propositions as types translation.

In chapter 3 we introduced propositional logic with the connectives  $\_ \wedge \_$  (and),  $\_ \vee \_$  (or), implication  $\_ \Rightarrow \_$ , negation  $\neg \_$ , logical equivalence  $\_ \Leftrightarrow \_$  and the logical constants **True** and **False**. Now we are going to extend this to predicate logic and we are going to introduce the quantifiers  $\forall$  (for all) and  $\exists$  (exists).

### 6.1 Predicates and Quantifiers

What is a predicate? Actually predicate is just another word for property. An example is the predicate  $\text{Even} : \mathbb{N} \rightarrow \text{prop}$ , a predicate is a function from some domain to propositions. Since we assume  $\text{prop} = \text{Set}$  a predicate is nothing but a dependent type. So for example  $\text{Even } 2 : \text{prop}$  is the proposition that 2 is even, while  $\text{Even } 3 : \text{prop}$  is the proposition that 3 is even. We expect that the former is non-empty, i.e. has evidence, while the should be no evidence for the latter.

Predicates with several inputs are called relations. Using currying we can represent relations without using products ( $\_ \times \_$ ) so for example

$$\_ \leq \_ : \mathbb{N} \rightarrow \mathbb{N} \rightarrow \text{prop}$$

is the less or equal relation. This is not limited to binary relations, an example for a ternary one is congruence modulo  $\_ \equiv \_ \text{ mod } \_$  e.g.  $3 \equiv 7 \text{ mod } 2 : \text{prop}$

expresses the (true) proposition that 3 and 7 have the same remainder when divided by 2.

To avoid repetition we just assume as given  $A : \text{Set}$  predicates  $P Q : A \rightarrow \text{prop}$  and a proposition  $R : \text{prop}$ .

We need to define the following operations:

$\forall [x \in A] P x$  For all  $x:A$ ,  $P x$  holds.

$\exists [x \in A] P x$  There is an  $x : A$ , such that  $P x$  holds.

So for example  $\forall [x \in \mathbb{N}] \text{Even } x \vee \text{Odd } x$  expresses that all natural are even or odd, and  $\exists [x \in \mathbb{N}] \text{Even } x$  expresses that there is an even natural number.

What is evidence for  $\forall [x \in A] P x$ ? It is a function that assigns to every  $x : A$  evidence for  $P x$ . Hence we use the dependent function type and define

$$\forall [x \in A] P x = (x : A) \rightarrow P x$$

What is evidence for  $\exists P x \in A, P x$ ? It is a pair of an element  $a : A$  and evidence for  $P a$ . Hence we use a dependent pair type or  $\Sigma$ -type to interpret exists:

$$\exists [x \in A] P x = \Sigma [x \in A] P x$$

We can use function application to instantiate a universally quantified assumption. That is given  $h : \forall [x \in A] P x$  and  $a : A$  we obtain  $f a : P a$ . To prove a universally quantified proposition  $\forall [x \in A] P x$  we assume as given an arbitrary  $x : A$  and prove  $p : P x$  and now  $\lambda x \rightarrow p : \forall [x \in A] P x$ .

To prove an existential statement  $\exists [x \in A] P x$  we need to find a witness  $a : A$  and show that the witness satisfies the property  $p : P a$  then  $(a, p) : \exists [x \in A] P x$ . On the other hand we can show the elimination principle for existential quantification: that is if we assume  $h : \exists [x \in A] P x$  and we can show that a given any  $x : A$  which that satisfies  $p x$  we can prove a proposition  $R$  then we know  $R$ . This can be expressed as follows:

$$\text{ex-e} : (\exists [x \in A] P x) \rightarrow (\forall [x \in A] (P x \Rightarrow R)) \rightarrow R$$

This is easy to prove using pattern matching:

$$\text{ex-e } (a, p) f = f a p$$

I noticed that students often struggle with this principle because it is a bit hard to verbalize. Here is my attempt using a concrete example” I use  $A$  to be the students in my class, I use  $P x$  to mean that  $x$  is clever and  $R$  means that I m happy. Now we know it is true that if any student is clever then I am happy this corresponds to  $\forall [x \in A] P x \rightarrow R$ .

We can now complete the table showing the propositions as types explanation for all logical connectives:

<i>latin</i>	<i>english</i>	<i>logic</i>	<i>types / definitions</i>
Conjunction	And	$P \wedge Q$	$P \times Q$
Disjunction	Or	$P \vee Q$	$P \uplus Q$
Implication	if-then	$P \Rightarrow Q$	$P \rightarrow Q$
Verum	true	True	$\top$
Falsum	false	False	$\perp$
Negation	not	$\neg P$	$P \Rightarrow \text{False}$
Equivalence	if-and-only-if	$P \Leftrightarrow Q$	$(P \Rightarrow Q) \wedge (Q \Rightarrow P)$
Universal quantification	for all	$\forall [x \in A] P x$	$(x : A) \rightarrow P x$
Existential quantification	exists	$\exists [x \in A] P x$	$\Sigma [x \in A] P x$

Abstract explanations like this are either obvious or incomprehensible. Hence the best is to look at some examples which is what we are going to do in the rest of this chapter.

## 6.2 Some Tautologies

To illustrate what we can do with our definition of quantifiers, we shall prove some tautologies, that is statements in predicate logic which are true in general.

The first tautology I want to prove is that  $\forall$  commutes with  $\wedge$ , that is

$$\text{taut}_1 : (\forall [x \in A] P x \wedge Q x) \Leftrightarrow (\forall [x \in A] P x) \wedge (\forall [x \in A] Q x)$$

I think it is always helpful to think of a concrete example: so let's say  $A$  is the set of students in the lecture,  $P x$  means that student  $x$  has yellow laces, and  $Q x$  means that  $x$  wears a black shirt. Now it is clearly the same to say *Everybody has yellow laces and wears a black shirt* and *Everybody has yellow laces and everybody wears a black shirt*, and this has nothing to do with students, laces or shirt colors.

Ok two prove the statement we need to prove two implications:

$$\begin{aligned} \text{proj}_1 \text{taut}_1 &= ? \\ \text{proj}_2 \text{taut}_1 &= ? \end{aligned}$$

Each of them is an implication, hence we may as well introduce parameters.

$$\begin{aligned} \text{proj}_1 \text{taut}_1 h &= ? \\ \text{proj}_2 \text{taut}_1 h &= ? \end{aligned}$$

In the first case we have an assumption  $h : \forall [x \in A] P x \wedge Q x$ , i.e. a dependent function and we need to prove  $(\forall [x \in A] P x) \wedge (\forall [x \in A] Q x)$ , that is a pair of functions  $?, ?$ . In the first component we need to prove  $\forall [x \in A] P x$  that is a function  $\lambda x \rightarrow ?$  where  $? : P x$ . Now we can derive  $h x : P x \wedge Q x$  and hence obtain  $\text{proj}_1 (h x) : P x$ . Similarly, we use  $\lambda x \rightarrow \text{proj}_2 (h x)$  for the 2nd component.

$$\text{proj}_1 \text{taut}_1 h = (\lambda x \rightarrow \text{proj}_1 (h x)), \lambda x \rightarrow \text{proj}_2 (h x)$$

For the other direction

$$\text{proj}_2 \text{taut}_1 h = ?$$

We have that  $h : (\forall [x \in A] P x) \wedge (\forall [x \in A] Q x)$  and we need to show  $? : \forall [x \in A] P x \wedge Q x$ . We first split our assumption  $h$  into two

$$\text{proj}_2 \text{taut}_1 (hp, hq) = ?$$

With  $hp : \forall [x \in A] P x$  and  $hq : \forall [x \in A] Q x$ . Now to prove  $? : \forall [x \in A] P x \wedge Q x$  we need to derive a function, hence we use  $\lambda x \rightarrow ?$  where  $? : P x \wedge Q x$ . Hence we need to produce a pair  $?, ? : P x \wedge Q x$  which we can do using  $hp x : P x$  and  $hq x : Q x$ . Hence we have

$$\text{proj}_2 \text{taut}_1 (hp, hq) = \lambda x \rightarrow hp x, hq x$$

What happens if we replace  $\_ \wedge \_$  by  $\_ \vee \_$  in the previous statement. Can we prove

$$(\forall [x \in A] P x \vee Q x) \Leftrightarrow (\forall [x \in A] P x) \vee (\forall [x \in A] Q x)$$

A quick analysis using common sense shows that this doesn't hold. It maybe that all students have yellow laces or black shirts but this doesn't mean that all students have yellow laces or all students have black shirts. I.e. the left to right implication doesn't hold. However, the right to left one is actually valid and can be proven:

$$\begin{aligned} \text{taut}_2 : (\forall [x \in A] P x) \vee (\forall [x \in A] Q x) &\Rightarrow (\forall [x \in A] P x \vee Q x) \\ \text{taut}_2 (\text{inj}_1 p) &= \lambda x \rightarrow \text{inj}_1 (p x) \\ \text{taut}_2 (\text{inj}_2 q) &= \lambda x \rightarrow \text{inj}_2 (q x) \end{aligned}$$

Ok, what happens if we start again with the first tautology but this time we replace  $\forall P$  with  $\exists P$ ? Does exists commute with and?

$$(\exists [x \in A] P x \wedge Q x) \Leftrightarrow (\exists [x \in A] P x) \wedge (\exists [x \in A] Q x)$$

Again common sense shows that this doesn't hold, it may well be that there is a student who has yellow laces and another one with a black shirt but this doesn't mean that there is one student with yellow laces and a black shirt. This time the implication from right to left fails but we can prove the one from left to right:

$$\begin{aligned} \text{taut}_3 : (\exists [x \in A] P x \wedge Q x) &\Rightarrow (\exists [x \in A] P x) \wedge (\exists [x \in A] Q x) \\ \text{taut}_3 (a, (p, q)) &= (a, p), (a, q) \end{aligned}$$

The code shows clearly why we can't turn it around. We use the same  $a$  twice but this won't work in the other direction when both witnesses may be different.

What happens if we do both replacements at the same time. Can we prove:

$$(\exists[x \in A] P x \vee Q x) \Leftrightarrow (\exists[x \in A] P x) \vee (\exists[x \in A] Q x)$$

And indeed we can:

$$\begin{aligned} \text{taut}_4 & : (\exists[x \in A] P x \vee Q x) \Leftrightarrow (\exists[x \in A] P x) \vee (\exists[x \in A] Q x) \\ \text{proj}_1 \text{taut}_4 (a, \text{inj}_1 p) & = \text{inj}_1 (a, p) \\ \text{proj}_1 \text{taut}_4 (a, \text{inj}_2 q) & = \text{inj}_2 (a, q) \\ \text{proj}_2 \text{taut}_4 (\text{inj}_1 (a, p)) & = a, \text{inj}_1 p \\ \text{proj}_2 \text{taut}_4 (\text{inj}_2 (a, q)) & = a, \text{inj}_2 q \end{aligned}$$

I leave it to you to go through the code line by line.

To summarize we can show:

$$\begin{aligned} (\forall[x \in A] P x \wedge Q x) & \Leftrightarrow (\forall[x \in A] P x) \wedge (\forall[x \in A] Q x) \\ (\exists[x \in A] P x \vee Q x) & \Leftrightarrow (\exists[x \in A] P x) \vee (\exists[x \in A] Q x) \end{aligned}$$

what is the pattern?  $\forall$  is represented as a  $\Pi$ -type which is an infinite product and  $\wedge$  is represented as  $\times$  which is binary product. Now it should be no surprise that products commute with products. In the same vein  $\exists$  was represented as a  $\Sigma$ -type which is an infinite sum and  $\vee$  was represented using  $\uplus$  which is a binary sum and again sums commute with sums.

## 6.3 Equality

While we usually define relations that are specific to a set (e.g.  $\_ \leq \_ : \mathbb{N} \rightarrow \mathbb{N} \rightarrow \text{prop}$ ), there is one relation which is used for all sets: equality. We write

$$\_ \equiv \_ : A \rightarrow A \rightarrow \text{prop}$$

That is given a  $b : A$  we write  $a \equiv b : \text{prop}$  for the proposition that  $a$  is equal to  $b$ . So for example we write  $1 + 2 \equiv 1 + 2 : \text{prop}$  which is actually inhabited, while  $0 \equiv 1 : \text{prop}$  is empty.

Equality is defined as an inductive type with one constructor: reflexivity.

```
data _≡_ : A → A → prop where
  refl : {a : A} → a ≡ a
```

Indeed, this is not fundamentally different to the dependent types we have already seen e.g. `Vec` and `Fin`. However, the special thing about equality is that the same index is repeated in the codomain of `refl`.

We can understand the equality type as saying that identical things are equal. This sounds trivial but before we weren't able to talk about identity. The equality type reflects the notion of identity as a type and hence we can talk about it, i.e. use it in a proposition.

Now that we have defined equality, we should verify that basic properties hold. What are the basic properties of equality? First of all it should be an

equivalence relation. A relation  $\_ \sim \_ : A \rightarrow A \rightarrow \text{prop}$  is an equivalence relation, if it is reflexive, symmetric and transitive. We can express this using a record:<sup>1</sup>

```
record isEqRel ( $\_ \sim \_ : A \rightarrow A \rightarrow \text{prop}$ ) : prop where
  field
    reflexive  : {a : A}      → a ~ a
    symmetric : {a b : A}    → a ~ b → b ~ a
    transitive : {a b c : A} → a ~ b → b ~ c → a ~ c
```

Ok let's show that  $\_ \equiv \_ \{A\} : A \rightarrow A \rightarrow \text{prop}$  is an equivalence relation. The first property, reflexivity, is obvious because it is exactly what the constructor `refl` says. What about symmetry?

```
sym : {a b : A} → a ≡ b → b ≡ a
```

```
sym h = ?
```

We have assumed  $a\ b : A$  and  $h : a \equiv b$  and our goal is to prove  $b \equiv a$ . We are going to use the fact that the only proof of  $a \equiv b$  is `refl` and pattern match on `h`. We end up with

```
sym refl = ?
```

And now we only have one assumption  $a : A$  and our goal is to prove  $a \equiv a$  because once we matched `h` with `refl` we also need to identify `a` and `b`. Now this is easy to complete:

```
sym refl = refl
```

To summarise: we can show that equality is symmetric because the only way we can prove  $a \equiv b$  is by `refl` and in this case `a` and `b` are identical and we can use `refl` to prove  $b \equiv a$  which is just  $a \equiv a$ .

Transitivity is also straightforward using basically the same idea:

```
trans : {a b c : A} → a ≡ b → b ≡ c → a ≡ c
```

Ok we start with:

```
trans p q = ?
```

where  $a\ b\ c : A$ ,  $p : a \equiv b$  and  $q : b \equiv c$  and we need to prove  $a \equiv c$ . An obvious move is to pattern match one of the equality proofs. But which one? It turns out it doesn't matter. Let's just go with `p`.

<sup>1</sup>Ok, at this point I am giving up on using the special  $\forall$ -syntax which translates into a dependent function type because it is more convenient to use  $\{.\}$   $\rightarrow$  here because we don't always want to fill in these parameters when using `sym` or any of the other combinators for equality.

trans refl q = ?

Matching `p` with `refl` has identified `a` and `b`. Hence we now only have `a c : A` and `q : a ≡ c` and our goal is still `a ≡ c` which just follow by assumption:

trans refl q = q

To summarise: we have shown that `_≡_ {A}` is an equivalence relation:

```
≡-is-eqRel : isEqRel (_≡_ {A})
≡-is-eqRel = record {
  reflexive  = refl;
  symmetric = sym;
  transitive = trans}
```

There are other properties of equality which are useful in proofs. Every function preserves equality, we say equality is a congruence.

cong : (f : A → B) → {a b : A} → a ≡ b → f a ≡ f b

Again this is easy to prove by pattern matching `a ≡ b`:

cong f refl = refl

At this point we may wonder why we can't prove the inverse of `cong`

uncong : (f : A → B) → {a b : A} → f a ≡ f b → a ≡ b

However, we cannot pattern match `f a ≡ f b`. Even though we know that it has to be proven by `refl` we don't know that the argument would have to be `f a`, and indeed `f` may be a constant function and there is no reason why we should assume that both sides need to be `f a`. Actually `uncong` states that every function is injective which clearly isn't true.

We can only pattern match an equality assumption if at least one side is a variable which can be identified with the other side — which implies it cannot contain this variable as a subterm.

Another important property of equality is that we can replace equals by equals. That is if we have a goal `P b` and we know that `a ≡ b` then we can prove `P a` instead. This is expressed by the following function:

subst : (P : A → prop) → {a b : A} → a ≡ b → P a → P b

which again is easy to prove using pattern matching:

subst P refl p = p

Indeed, all the principles we have proved before can be easily derived from `subst`. I'll leave this as an exercise.

### 6.3.1 Propositional vs definitional equality

Sometimes it is mentioned as a criticism about Type Theory that we have two equalities. And indeed, maybe you have noticed that I sometimes use  $\_ = \_$  to talk about equalities which follow directly from the definitions of functions and now we have introduced  $\_ \equiv \_$  as *propositional equality*. Isn't this rather confusing.

I don't think so. There is only one equality we can talk about *in* Type Theory and this is exactly propositional equality  $\_ \equiv \_$ , which captures precisely what we mean by equality in Mathematics. However, when we talk *about* Type Theory then Type Theory is indeed a bit more involved than conventional logic, in that we also have a static equality, i.e. definitional equality. And we cannot talk about this in Type Theory as we cannot talk about the inhabitation  $\_ : \_$ . Both are called *judgements* and not propositions. They are static properties of the language, so to say rules of the grammar but not the content we are talking about.

Once we accept that  $a : A$  is a judgement we also need to be able to talk about a judgemental equality. For example let's say I define

$$\begin{aligned} n &: \mathbb{N} \\ n &= 3 \end{aligned}$$

and later I construct:

$$\begin{aligned} v &: \text{Vec } \mathbb{N} \ n \\ v &= 1 :: 2 :: 3 :: [] \end{aligned}$$

Then to see that the definition of  $v$  is well formed, i.e. is statically an acceptable text we need to know statically that  $n = 3$ . Which means that equality is a part of the judgement that a certain text is grammatically correct.

Now when I present Type Theory, i.e. in a book about Type Theory or in a lecture I am often switching between two modes: I am talking about Type Theory and then I am talking in Type Theory. Hence I will sometimes mention that certain expressions are definitionally equal using  $\_ = \_$  even though this is nothing I can talk about in Type Theory. And certainly if two expressions are definitionally equal then the values they denote are indeed propositionally equal just using *refl*.

## 6.4 Proving properties of addition

Ok, now we have the tools, let's do something with them. Actually let's build more tools. When reasoning about numbers we need to use algebra and what are the most basic laws of algebra? Here are two basic laws about addition on natural numbers: adding zero doesn't change anything. Why two laws? Since we haven't yet proven that  $x + y = y + x$  we can add zero on the left  $0 + n \equiv n$  or on the right  $n + 0 \equiv n$ .

Before we look at them let's recall the definition of addition from section 4.1.2:

$$\begin{aligned} \_ + \_ &: \mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N} \\ \text{zero} + n &= n \\ \text{suc } m + n &= \text{suc } (m + n) \end{aligned}$$

Now, let's prove the laws. Starting with adding zero on the left:

$$\text{lneutr+} : \{n : \mathbb{N}\} \rightarrow 0 + n \equiv n$$

It turns out that this law is trivial, because this follows directly from our definition of  $\_ + \_$ . Hence we just need to say

$$\text{lneutr+} = \text{refl}$$

and we are done.

The other direction is less straightforward but more interesting.

$$\text{rneutr+} : \{n : \mathbb{N}\} \rightarrow n + 0 \equiv n$$

We cannot say  $\text{rneutr+} = \text{refl}$  because the two sides are not identical or compute to the same expression. However, we can still prove it by using recursion. For this purpose first of all let's make the parameter visible

$$\text{rneutr+ } \{n\} = ?$$

Ok, we have assumed  $n : \mathbb{N}$  what can we do with it? Exactly, we can pattern match on it:

$$\begin{aligned} \text{rneutr+ } \{\text{zero}\} &= ? \\ \text{rneutr+ } \{\text{suc } n\} &= ? \end{aligned}$$

The first goal is  $\text{zero} + \text{zero} \equiv \text{zero}$  and this is easy to prove again, because we know from the definition of  $\_ + \_$  that  $\text{zero} + n = n$ , hence

$$\text{rneutr+ } \{\text{zero}\} = \text{refl}$$

In the  $\text{suc } n$  case our goal is  $\text{suc } n + \text{zero} \equiv \text{suc } n$ . While here we can't reduce to identical expressions we can compute a bit exploiting the second line in the definition of  $\_ + \_$  which tells us that  $\text{suc } m + n \equiv \text{suc } (m + n)$ . Applying this to the left side of our goal it simplifies to  $\text{suc } (n + \text{zero}) \equiv \text{suc } n$ . This suggests we could recursively prove  $n + \text{zero} \equiv n$  but we need to get rid of the pesky  $\text{suc}$ . But we can, we just need to use  $\text{cong suc}$ :

$$\text{rneutr+ } \{\text{suc } n\} = \text{cong suc } ?$$

And now indeed the goal is  $n + \text{zero} \equiv n$  and we can use a recursive call:

$$\text{rneutr+ } \{\text{suc } n\} = \text{cong suc } (\text{rneutr+ } \{n\})$$

What has actually happened? We just used recursion in a proof! And it meant that to prove that some property holds for all natural numbers it is enough to prove it for zero and prove it for  $\text{suc } n$  using that it holds for  $n$ . This proof principle has a name, it is called *the principle of induction*. This is a nice consequence of propositions as types: induction is recursion.

Let's move on. Another very useful property of addition is that it doesn't matter how we bracket several additions, that is:

$$\text{assoc+} : \{l \ m \ n : \mathbb{N}\} \rightarrow (l + m) + n \equiv l + (m + n)$$

This property is called *associativity*. Again our strategy is to use pattern matching and recursion but this time we have three numbers to choose from:  $l$ ,  $m$  and  $n$ . It turns out that the first choice  $l$  is the right one, but why? Looking at the definition of addition we notice that an expression of the form  $e_1 + e_2$  computes if  $e_1$  starts with a constructor. Hence it is a good idea to turn expressions on the left into constructors and  $l$  is best positioned to achieve this.

$$\begin{aligned} \text{assoc+ } \{\text{zero}\} &= ? \\ \text{assoc+ } \{\text{suc } l\} &= ? \end{aligned}$$

The first goal is to show  $(\text{zero} + m) + n \equiv \text{zero} + (m + n)$  and using the definition of  $\_+\_$  we can see

$$\begin{aligned} (\text{zero} + m) + n &= m + n \\ \text{zero} + (m + n) &= m + n \end{aligned}$$

Hence we just need to say:

$$\text{assoc+ } \{\text{zero}\} = \text{refl}$$

Ok, what about the  $\text{suc } l$  case? We need to prove  $(\text{suc } l + m) + n \equiv \text{suc } l + (m + n)$ . Again we can use the definition of  $\_+\_$ :

$$\begin{aligned} (\text{suc } l + m) + n &= \text{suc } (l + m) + n \\ &= \text{suc } ((l + m) + n) \\ (\text{suc } l) + (m + n) &= \text{suc } (l + (m + n)) \end{aligned}$$

Hence all we need to do is to use  $\text{cong suc}$  and then we can use recursion:

$$\text{assoc+ } \{\text{suc } l\} = \text{cong suc } (\text{assoc+ } \{l\})$$

In algebra there is a name for an operation (like  $\_+\_$ ) and an element which behaves like 0: it is called *a monoid*:

**record** IsMonoid  
 $(e : A) (\_ \bullet \_ : A \rightarrow A \rightarrow A) : \text{prop where}$

**field**

$$\begin{aligned} \text{lneutr} &: \{x : A\} \rightarrow e \bullet x \equiv x \\ \text{rneutr} &: \{x : A\} \rightarrow x \bullet e \equiv x \\ \text{assoc} &: \{x y z : A\} \rightarrow (x \bullet y) \bullet z \equiv x \bullet (y \bullet z) \end{aligned}$$

and we can summarize our results so far:

$$\begin{aligned} +\text{isMonoid} &: \text{IsMonoid zero } \_ + \_ \\ +\text{isMonoid} &= \mathbf{record} \{ \\ &\text{lneutr} = \text{lneutr+}; \\ &\text{rneutr} = \text{rneutr+}; \\ &\text{assoc} = \lambda \{x\} \{y\} \{z\} \rightarrow \text{assoc+ } \{x\} \{y\} \{z\} \} \end{aligned}$$

However, addition is one more useful property: it is commutative  $m + n \equiv n + m$ . How can we prove this?

$$\text{comm+} : \{m n : \mathbb{N}\} \rightarrow m + n \equiv n + m$$

Again the question is over which argument to do the induction (aka recursion). However, due to the symmetry of the equation it doesn't matter - hence let's use the first.

$$\begin{aligned} \text{comm+ } \{\text{zero}\} \{n\} &= ? \\ \text{comm+ } \{\text{suc } m\} \{n\} &= ? \end{aligned}$$

We get two goals:

$$\begin{aligned} \text{zero} + n &\equiv n + \text{zero} \\ \text{suc } m + n &\equiv n + \text{suc } m \end{aligned}$$

In both cases we can simplify the left hand side, but there is no simplification possible for the righthand sides, because they are stuck on  $n$ .

$$\begin{aligned} n &\equiv n + \text{zero} \\ \text{suc } (m + n) &\equiv n + \text{suc } m \end{aligned}$$

The first goal can be handled with  $\text{rneutr+}$  but it is not clear how to make progress with the 2nd. We need a way to move the successor from the 2nd argument to the front - while the definition only provides a way to move the successor from inside the first argument. Hence we suspend this proof and first prove a lemma:

$$\text{suc+} : \{m n : \mathbb{N}\} \rightarrow \text{suc } (n + m) \equiv n + \text{suc } m$$

In this case it is clear that the induction should use  $n$ . The rest is straight forward:

$$\begin{aligned} \text{suc+ } \{m\} \{\text{zero}\} &= \text{refl} \\ \text{suc+ } \{m\} \{\text{suc } n\} &= \text{cong suc } (\text{suc+ } \{m\} \{n\}) \end{aligned}$$

Back to our original problem:

$$\text{comm+} : \{m\ n : \mathbb{N}\} \rightarrow m + n \equiv n + m$$

While as remarked the zero case can be handled with `renutr+` there is a slight inconvenience that the equation is the wrong way around: `rneutr` proves  $n + \text{zero} \equiv n$  but we need  $n \equiv n + \text{zero}$ . This can be easily dealt with using `sym`:

$$\text{comm+} \{ \text{zero} \} \{ n \} = \text{sym rneutr+}$$

For the second case we have to combine the lemma `suc+` we have shown and the recursive call. That is to show

$$\text{suc } m + n \equiv n + \text{suc } m$$

We use

$$\begin{aligned} \text{suc } m + n &= \text{suc } (m + n) && \text{The definition of } + \\ &= \text{suc } (n + m) && \text{The induction hypothesis for } m \\ &= n + \text{suc } m && \text{The lemma } \text{suc+} \end{aligned}$$

Using `trans` and `cong` we can translate this into agda code

$$\text{comm+} \{ \text{suc } m \} \{ n \} = \text{trans } (\text{cong } \text{suc } (\text{comm+} \{ m \} \{ n \})) \text{suc+}$$

It is clear that this presentation of equational proofs will become very unreadable very quickly. The standard agda library defines some syntactic sugar which provides a nicer presentation which shows the intermediate goals and enables us to layout the derivation nicely but which actually reduces to the same proof.

```

comm+ {suc m} {n} = begin
  suc m + n
  ≡⟨ refl ⟩
  suc (m + n)
  ≡⟨ cong suc (comm+ {m} {n}) ⟩
  suc (n + m)
  ≡⟨ suc+ ⟩
  n + suc m
  ■

```

However, we still have to spend a lot of space for mathematically trivial steps. This is can be addressed using a solver which automatically computes solutions.

To summarise we have shown that `_+_` forms a *commutative monoid*:

```

record IsCommMonoid
  (e : A) (_•_ : A → A → A) : prop where
  field

```

```

isMonoid : IsMonoid e _•_
isComm   : {x y : A} → x • y ≡ y • x

+isCommMonoid : IsCommMonoid zero _+_
+isCommMonoid = record {
  isMonoid = +isMonoid;
  isComm   = λ {x} {y} → comm+ {x} {y}}

```

## 6.5 Extending the negative translation

We have introduced the negative translation in section 3.4 as an alternative to the boolean interpretation of classical logic. Now moving to predicate logic we cannot actually use the boolean interpretation because given  $P : A \rightarrow \text{Bool}$  how can we determine a boolean to  $\forall [x : A] P x$ . In particular it may be that  $A$  is infinite, e.g.  $A = \mathbb{N}$ .

However, the negative translation which translates any proposition  $P$  so that  $\text{RAA } P = \neg (\neg P) \rightarrow P$  holds still works as we will see. First of all we need to show that  $\text{RAA}$  holds for  $\forall [x \in A] P x$  if for all  $x : A$  we have that  $\text{RAA } (P x)$  holds.

```

all-neg : ((x : A) → RAA (P x)) → RAA (∀ [x ∈ A] P x)
all-neg raa h x = raa x (λ g → h (λ k → g (k x)))

```

What about  $\exists [x \in A] P x$ ? Like  $P \vee Q$  this is a positive statement, i.e. it contains information, namely the choice of witness. Hence as for  $\_ \vee \_$  we translate  $\exists$  using the classical equivalence that  $\exists [x \in A] P x$  is equivalent to saying that it is not the case the property isn't false for all  $x : A$ .

```

Ex-c : (A : Set) (P : A → prop) → prop
Ex-c A P = ¬ (∀ [x ∈ A] ¬ (P x))

```

I am going to write  $\exists^c [x \in A] P$  for classical existence. As for  $\_ \vee^c \_$ . It should be clear that we have:

```

neg-ex-c : RAA (∃c [x ∈ A] P)

```

because we have already shown that all negated propositions satisfy  $\text{RAA}$ .

We need to show that the basic principles for existentials are provable, that is we need to show:

```

exc-i : (a : A) → P a → ∃c [x ∈ A] P x
exc-e : RAA R → ((x : A) → P x → R) → ∃c [x ∈ A] P x → R

```

What about equality? It turns out that most equalities we encounter are already negative, for example  $\_ \equiv \_ \{ \mathbb{N} \}$  satisfies  $\text{RAA}$  because it is decidable as

we are going to show in the next chapter. However, we don't need to assume this but we can define classical equality as the double negation of the usual equality:

$$\begin{aligned} \_ \equiv^c \_ &: A \rightarrow A \rightarrow \text{prop} \\ \mathbf{a} \equiv^c \mathbf{b} &= \neg(\neg(a \equiv b)) \end{aligned}$$

and as before we have:

$$\text{ne-eq-c} : \{a \ b : A\} \rightarrow \text{RAA } (a \equiv^c b)$$

We can derive some the basic principles associated with equality for  $\_ \equiv^c \_$ :

$$\begin{aligned} \text{refl}^c &: \{a : A\} \rightarrow a \equiv^c a \\ \text{subst}^c &: (P : A \rightarrow \text{prop}) \rightarrow (\{x : A\} \rightarrow \text{RAA } (P \ x)) \\ &\rightarrow \{a \ b : A\} \rightarrow a \equiv^c b \rightarrow P \ a \rightarrow P \ b \end{aligned}$$

I am leaving the verification of these properties as an exercise.

Hence it seems that classical logic is just the logic of the negative fragment of intuitionistic logic where we use negative, i.e. classical versions, of disjunction and existence. Almost, but there is another important aspect of classical Mathematics: the axiom of choice which we are going to discuss later. The negative translation is not going to help us here because the negative translation of the axiom of choice is not provable intuitionistically.

## 6.6 History

An early form of predicate logic was introduced by Frege in the late 19th century [Fre79]. His basic laws allowed quantification over all sets and hence was inconsistent as Russell pointed out. However, his attempt to make the rules of reasoning precise lead to the formal system of predicate logic. This is also often called first order logic, because we don't allow quantification over propositions themselves.

While Frege tried to capture set theory and logic at the same time, later it became clear that it is a good idea to separate the system of reasoning and the particular basic assumptions, i.e. axioms. Hence for example Peano Arithmetic and Zermelo-Fraenkel set theory can be formulated in the same framework. It was Hilbert and Ackermann who clearly formulated first order logic in the 1930ies, (there is a recent english translation: [HA99]).

Gödel proved a completeness theorem for classical predicate logic which shows that the formal system is complete for the classical notion of a model, i.e. if a proposition holds in all models it is also provable [Kur29]. However, this is done for classical logic in a classical metatheory and in this form it is not relevant in our context.

Unlike set theory type theory doesn't use the framework of predicate logic it stands on its own. However, many of the idioms of predicate logic are used and

we can understand them via their translation into predicate logic as we have shown. However, the interpretation of disjunction  $\vee$  and existence  $\exists$  we have represented are debatable since the types don't really behave like propositions. We will return to this issue later and present an alternative view what is a proposition.

## 6.7 Exercises

1. We are playing logic poker again. But this time with predicate logic!

Consider the following propositions:

$$\begin{aligned}
P01 &= \{A\ B : \text{Set}\} \{R : A \rightarrow B \rightarrow \text{prop}\} \rightarrow \\
&\quad ((\forall [x \in A] \exists [y \in B] R\ x\ y)) \rightarrow \\
&\quad (\exists [y \in B] \forall [x \in A] R\ x\ y) \\
P02 &= \{A\ B : \text{Set}\} \{R : A \rightarrow B \rightarrow \text{prop}\} \rightarrow \\
&\quad (\exists [y \in B] (\forall [x \in A] R\ x\ y)) \\
&\quad \rightarrow (\forall [x \in A] (\exists [y \in B] R\ x\ y)) \\
P03 &= \{A : \text{Set}\} \{P : A \rightarrow \text{prop}\} \rightarrow \\
&\quad \neg (\exists [x \in A] P\ x) \rightarrow \forall [x \in A] \neg (P\ x) \\
P04 &= \{A : \text{Set}\} \{P : A \rightarrow \text{prop}\} \rightarrow \\
&\quad (\forall [x \in A] \neg (P\ x)) \rightarrow \neg (\exists [x \in A] P\ x) \\
P05 &= \{A : \text{Set}\} \{P : A \rightarrow \text{prop}\} \rightarrow \\
&\quad (\neg (\forall [x \in A] P\ x)) \rightarrow \exists [x \in A] \neg (P\ x) \\
P06 &= \{A : \text{Set}\} \{P : A \rightarrow \text{prop}\} \rightarrow \\
&\quad (\exists [x \in A] \neg (P\ x)) \rightarrow (\neg (\forall [x \in A] P\ x)) \\
P07 &= \{A : \text{Set}\} \{P : A \rightarrow \text{prop}\} \rightarrow \\
&\quad (\neg (\neg (\forall [x \in A] P\ x))) \rightarrow \forall [x \in A] \neg (\neg (P\ x)) \\
P08 &= \{A : \text{Set}\} \{P : A \rightarrow \text{prop}\} \rightarrow \\
&\quad (\forall [x \in A] \neg (\neg (P\ x))) \rightarrow (\neg (\neg (\forall [x \in A] P\ x))) \\
P09 &= \{A : \text{Set}\} \{P : A \rightarrow \text{prop}\} \rightarrow \\
&\quad (\exists [x \in A] \top) \rightarrow (\exists [x \in A] P\ x) \rightarrow \forall [x \in A] P\ x \\
P10 &= \{A : \text{Set}\} \{P : A \rightarrow \text{prop}\} \rightarrow \\
&\quad (\exists [x \in A] \top) \rightarrow (\exists [y \in A] (P\ y \rightarrow \forall [x \in A] P\ x))
\end{aligned}$$

Here is a quick summary of the rules of logic poker:

- (a) You can prove it directly (intuitionistically) - just provide  $pi : Pi$ .
- (b) You cannot prove it with intuitionistic logic but you can prove it with classical logic. In this case prove  $pi : \text{CLASS} \rightarrow Pi$  where

$$\text{CLASS} = \{P : \text{prop}\} \rightarrow \text{RAA } P$$

- (c) It isn't even true in classical logic. In this case prove  $pi : Pi \rightarrow \perp$ . Note that you will have to come up with predicates and relations to refute the proposition.

2. Show that all the properties of equality we have shown are derivable from `subst`. That is prove

$$\begin{aligned} \text{sym-s} &: \{a\ b : A\} \rightarrow a \equiv b \rightarrow b \equiv a \\ \text{trans-s} &: \{a\ b\ c : A\} \rightarrow a \equiv b \rightarrow b \equiv c \rightarrow a \equiv c \\ \text{cong-s} &: (f : A \rightarrow B) \rightarrow \{a\ b : A\} \rightarrow a \equiv b \rightarrow f\ a \equiv f\ b \end{aligned}$$

using only:

$$\text{subst} : (P : A \rightarrow \text{prop}) \rightarrow \{a\ b : A\} \rightarrow a \equiv b \rightarrow P\ a \rightarrow P\ b$$

instead of pattern matching.

3. Verify the laws of boolean algebra we discussed in section 3.1.

$$\begin{aligned} \text{com-}\& : \{x\ y : \text{Bool}\} \rightarrow x \& y \equiv y \& x \\ \text{distr-}\&| & : \{x\ y\ z : \text{Bool}\} \rightarrow x \& (y | z) \equiv x \& y | x \& z \\ \text{dm1} & : \{x\ y : \text{Bool}\} \rightarrow !(x | y) \equiv !x \& !y \\ \text{dm2} & : \{x\ y : \text{Bool}\} \rightarrow !(x \& y) \equiv !x | !y \end{aligned}$$

4. Show that the natural numbers  $\mathbb{N}$  with addition and multiplication form a commutative semiring. We have already shown that addition forms a commutative monoid, hence it is left to show:

$$\begin{aligned} \text{*isCommMonoid} & : \text{IsCommMonoid } 1 \ \_ \ * \ \_ \\ \text{null-l} & : \{n : \mathbb{N}\} \rightarrow 0 * n \equiv 0 \\ \text{distr-l} & : \{l\ m\ n : \mathbb{N}\} \rightarrow (l + m) * n \equiv l * n + m * n \end{aligned}$$

Note that I have mitted the symmetric forms of the distributivity laws:

$$\begin{aligned} \text{null-r} & : \{n : \mathbb{N}\} \rightarrow n * 0 \equiv 0 \\ \text{distr-r} & : \{l\ m\ n : \mathbb{N}\} \rightarrow l * (m + n) \equiv l * m + l * n \end{aligned}$$

since they follow from commutativity but it may be useful to prove them before establishing commutativity.

5. Continuing from the previous question: show that the natural numbers form an exponential semi-ring, that is prove the laws of high school algebra:

$$\begin{aligned} \text{exp-plus} & : \{l\ m\ n : \mathbb{N}\} \rightarrow l \uparrow (m + n) \equiv l \uparrow m * l \uparrow n \\ \text{exp-times} & : \{l\ m\ n : \mathbb{N}\} \rightarrow l \uparrow (m * n) \equiv (l \uparrow m) \uparrow n \end{aligned}$$

6. Prove the missing lemmas about the negative translation from section 6.5.

$$\begin{aligned} \text{all-neg} & : ((x : A) \rightarrow \text{RAA } (P\ x)) \rightarrow \text{RAA } (\forall [x \in A] P\ x) \\ \text{ex}^c\text{-i} & : (a : A) \rightarrow P\ a \rightarrow \exists^c [x \in A] P\ x \\ \text{ex}^c\text{-e} & : \text{RAA } R \rightarrow ((x : A) \rightarrow P\ x \rightarrow R) \rightarrow \exists^c [x \in A] P\ x \rightarrow R \\ \text{refl}^c & : \{a : A\} \rightarrow a \equiv^c a \\ \text{subst}^c & : (P : A \rightarrow \text{prop}) \rightarrow (\{x : A\} \rightarrow \text{RAA } (P\ x)) \\ & \rightarrow \{a\ b : A\} \rightarrow a \equiv^c b \rightarrow P\ a \rightarrow P\ b \end{aligned}$$